

```

/*****
The stata codes for the workshop on "Growth Curve Modeling in Stata
Feb, 12, 2018
*****/

cd "F:\workshop\HLM and SEM approaches to Growth Curve Modeling"

log using "groth curve in stata.log", replace

*****/
* Read in the data
*****/

use http://www.stata-press.com/data/r14/childweight.dta, clear

*****/
* Describe the data
*****/

des
list in 1/16, sepby(id)

*****/
* A Linear relation between age and weight, assuming no gender differences
*****/
graph twoway (scatter weight age) (lfit weight age),          xtitle(Age in years) ytitle(Weight in kg)
graph save regress1, replace

*****/
* Respective linear relations between weight and age for boys and girls
*****/
graph twoway (scatter weight age) (lfit weight age), by(girl) xtitle(Age in years) ytitle(Weight in kg)
graph save regress1_gender, replace

graph twoway (line weight age, connect(ascending)), by(girl) xtitle(Age in years) ytitle(Weight in kg)
graph save regress1_trajectory, replace

*****/
* Respective Linear growth curve for boys and girls
*****/
reg weight age if girl ==0
reg weight age if girl ==1

*****/
* Respective curvelinear growth curves for girls and boys, respectively
*****/
reg weight c.age##c.age if girl ==0
reg weight c.age##c.age if girl ==1

*****/
* Growth Curve Modeling
*****/

*****/
* Model_0 : Traditional regression
*****/

*****/
* Model_0: Regression commands
*****/

reg weight age
predict p_weight

graph twoway (line p_weight age, connect(ascending))
graph save model_0_0, replace

graph twoway (line p_weight age if girl ==0, connect(ascending))
graph save model_0_1, replace

graph twoway (line p_weight age if girl ==0, connect(ascending))
graph save model_0_2, replace

```

```
*****
* HLM commands
*****
mixed weight c.age, nolog
est store model_0

*****
* Model 1 : Linear Growth curve model with random intercept
*****

mixed weight age || id: , nolog
graph save model_1, replace
est store model_1

*****
* Model 2: Linear Growth curve model with random slope
*****

mixed weight age || id: c.age, covariance(unstructured) nolog
graph save model_2, replace
est store model_2

*****
* Model 3 : Curvilinear Growth model with random intercept
*****

mixed weight age c.age#c.age || id: c.age, covariance(unstructured) nolog
graph save model_3, replace
est store model_3

*****
* Compare Models 1 through 3
*****
lrtest model_0 model_1
lrtest model_1 model_2
lrtest model_2 model_3

*****
* Model 4: Same linear and curvilinear time effects for boys and girls
*****

mixed weight age c.age#c.age i.girl || id: c.age, covariance(unstructured) nolog

margins i.girl, at(age=(0(1)3)) vsquish
marginsplot, name(model_4, replace) x(age)

*****
* Model 5: Different linear and curvilinear time effects for boys and girls
*****

mixed weight i.girl##c.age##c.age || id: c.age, covariance(unstructured) nolog
est store model_5

margins i.girl, at(age=(0(1)3)) vsquish
marginsplot, name(model_5, replace) x(age)

*****
* close the log file
*****
log close
```